
1 Overview

This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2019-06_D20wk25 related to the title of this CR.

Contact:
Shaun Harry

shaun.harry@keysight.com
2 Corrections required
2.1 Change 1

	Function name
	f_NR5GC_RRC_Idle_Steps5_9_AKA()

	Reason for change
	The NAS counts are currently being reset at NAS EMU during Step 9Aa1 - 9Aa2 because the parameter p_NasCountReset in f_NR_NASSecurityModeProcedure() defaults to “true”

	Summary of change
	Pass p_NasCountReset as “omit” so that NAS EMU counts are not reset

	TTCN module
	NR5GC_NASSteps

	MCC160 Comment
	


Before Change

	  function f_NR5GC_RRC_Idle_Steps5_9_AKA (inout GMM_MobilityInfo_Type p_GMM_MobilityInfo_Type,

                                           NR_CellId_Type p_CellId,

                                           NG_NAS_MSG_Indication_Type p_ReceivedMsg,

                                           NAS_PlmnId p_PLMN,

                                           TypeOfRegistration_Type p_RegType := Initial_Secure,

                                           InterworkWithoutN26 p_IwkN26 := NOT_SUPPORTED)  runs on NR5GC_PTC return NasCount_Type

  {

    var template NAS_SecurityReturnType v_NAS_SecurityReturnValue;

    var NG_NAS_MSG_Indication_Type v_ReceivedMsg := p_ReceivedMsg;

    var NG_NAS_SecurityParams_Type v_SecurityParams := f_NR5GC_Security_Get();

    var SecurityHeaderType v_SentSecurityHeader;

    var template (present) SecurityHeaderType v_SecurityHeaderResponse;

    ………
    // Step 9Aa1 - 9Aa2

    if (p_RegType == Initial_NoSecurity or p_RegType == Initial_Secure) { // @sic R5-202564 sic@

      if (isvalue(p_GMM_MobilityInfo_Type.GMMCap)) {

        if (match('1'B, p_GMM_MobilityInfo_Type.GMMCap.s1Cap)  and p_IwkN26 == NOT_SUPPORTED) {

          v_NAS_SecurityReturnValue := f_NR_NASSecurityModeProcedure(p_CellId, v_SecurityParams, p_RegType, p_GMM_MobilityInfo_Type);
       }

      }

    }

    return valueof(v_NAS_SecurityReturnValue.NasCount);

  }


After Change

	  function f_NR5GC_RRC_Idle_Steps5_9_AKA (inout GMM_MobilityInfo_Type p_GMM_MobilityInfo_Type,

                                           NR_CellId_Type p_CellId,

                                           NG_NAS_MSG_Indication_Type p_ReceivedMsg,

                                           NAS_PlmnId p_PLMN,

                                           TypeOfRegistration_Type p_RegType := Initial_Secure,

                                           InterworkWithoutN26 p_IwkN26 := NOT_SUPPORTED)  runs on NR5GC_PTC return NasCount_Type

  {

    var template NAS_SecurityReturnType v_NAS_SecurityReturnValue;

    var NG_NAS_MSG_Indication_Type v_ReceivedMsg := p_ReceivedMsg;

    var NG_NAS_SecurityParams_Type v_SecurityParams := f_NR5GC_Security_Get();

    var SecurityHeaderType v_SentSecurityHeader;

    var template (present) SecurityHeaderType v_SecurityHeaderResponse;

    ……..
    // Step 9Aa1 - 9Aa2

    if (p_RegType == Initial_NoSecurity or p_RegType == Initial_Secure) { // @sic R5-202564 sic@

      if (isvalue(p_GMM_MobilityInfo_Type.GMMCap)) {

        if (match('1'B, p_GMM_MobilityInfo_Type.GMMCap.s1Cap)  and p_IwkN26 == NOT_SUPPORTED) {

          v_NAS_SecurityReturnValue := f_NR_NASSecurityModeProcedure(p_CellId, v_SecurityParams, p_RegType, p_GMM_MobilityInfo_Type, omit);//KS R5sxx change_IRAT_common_1

       }

      }

    }

    return valueof(v_NAS_SecurityReturnValue.NasCount);

  }


2.2 Change 2
	Function name
	f_NR5GC_IRAT_S1ToN1_IdleMode()

	Reason for change
	EUTRA PTC will alway send security info to NR before reselection to N1(in f_EUTRA38_Send_IRATParametersToNR()) which is not currently handled on the NR PTC.

	Summary of change
	Add handling for receiving security information from EUTRA before S1 to N1 reselection
Note : Existing commented code is reinstated

Note : if this security info is not needed in NR PTC then the alternative solution is to remove the sending in f_EUTRA38_Send_IRATParametersToNR()


	TTCN module
	NR5GC_IRAT_CommonFunctions

	MCC160 Comment
	


Before Change

	  function f_NR5GC_IRAT_S1ToN1_IdleMode(NR_CellId_Type p_CellId,

                                        RRCConnectionReleaseRequired_Type p_ReleaseConnection := rrcConnectionRelease,

                                        boolean p_FirstTimeS1ToN1 := true,

                                        template (omit) NG_MobileIdentity p_GUTIFromPreamble := omit,

                                        template NG_TrackingAreaId p_TAI := *,

                                        template (omit) NG_MobileIdentity p_GUTIToSend := omit,

                                        template (omit) NG_TrackingAreaIdList p_TAIList := omit,

                                        boolean p_Registered_Previously_5GCN := true)  runs on NR5GC_PTC

  {

    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (p_CellId);

    var PLMN_Identity v_ASN_PLMN := f_NR_CellInfo_GetPLMN (p_CellId);

    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(v_ASN_PLMN);

    var template (value) NG_MobileIdentity v_GUTIFromPreamble;

    var template NG_TrackingAreaId v_TAI;

    var RRCConnectionReleaseRequired_Type v_ReleaseConnection := p_ReleaseConnection;

    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;

    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(EUTRA);

 //   var EUTRA_NR_CoOrd_Security_Type v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (EUTRA);
    var NG_NAS_SecurityParams_Type v_NAS_SecurityParams := f_NR5GC_Security_Get();

    if (p_FirstTimeS1ToN1) { // If this is the 1st time then must release RRC Connection before re-registration and PDU session Modification

      v_ReleaseConnection := rrcConnectionRelease;

    }

    …….

  }


After Change

	function f_NR5GC_IRAT_S1ToN1_IdleMode(NR_CellId_Type p_CellId,

                                        RRCConnectionReleaseRequired_Type p_ReleaseConnection := rrcConnectionRelease,

                                        boolean p_FirstTimeS1ToN1 := true,

                                        template (omit) NG_MobileIdentity p_GUTIFromPreamble := omit,

                                        template NG_TrackingAreaId p_TAI := *,

                                        template (omit) NG_MobileIdentity p_GUTIToSend := omit,

                                        template (omit) NG_TrackingAreaIdList p_TAIList := omit,

                                        boolean p_Registered_Previously_5GCN := true)  runs on NR5GC_PTC

  {

    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (p_CellId);

    var PLMN_Identity v_ASN_PLMN := f_NR_CellInfo_GetPLMN (p_CellId);

    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(v_ASN_PLMN);

    var template (value) NG_MobileIdentity v_GUTIFromPreamble;

    var template NG_TrackingAreaId v_TAI;

    var RRCConnectionReleaseRequired_Type v_ReleaseConnection := p_ReleaseConnection;

    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;

    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(EUTRA);

    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (EUTRA);
    var NG_NAS_SecurityParams_Type v_NAS_SecurityParams := f_NR5GC_Security_Get();

    if (p_FirstTimeS1ToN1) { // If this is the 1st time then must release RRC Connection before re-registration and PDU session Modification

      v_ReleaseConnection := rrcConnectionRelease;

    }

    ……..

  }


2.3 Change 3
	Function name
	cr_TAU_Request_S1ToN1

	Reason for change
	In TRACKING AREA UPDATE REQUEST following S1->N1 reselecion, The Type of Security Context will be “Mapped security context” and the Active Flag will be “No bearer establishment requested”

	Summary of change
	Change expect TRACKING AREA UPDATE REQUEST contents

Note : An associated prose CR on 38.508-1 will be raised at RAN5#88

	TTCN module
	EUTRA38_NR5GCProcedures

	MCC160 Comment
	


Before Change

	template (present) NAS_UL_Message_Type cr_TAU_Request_S1ToN1(template (present)  EPS_UpdateTypeValue p_EpsUpdate_TypeValue,

                                                                NAS_KsiValue p_KsiValue,

                                                                template (present) MobileIdentity p_MobileId) :=

  { /* @status    APPROVED (NR5GC_IRAT) */

    tRACKING_AREA_UPDATE_REQUEST := {

      securityHeaderType         := tsc_SHT_NoSecurityProtection,

      protocolDiscriminator      := tsc_PD_EMM,

      messageType                := tsc_MT_TauRequest,

      asmeNasKeySetId            := cr_NAS_KeySetIdentifier(p_KsiValue, tsc_NasKsi_NativeSecurityContext),

      epsUpdateType              := cr_EPS_UpdateType(tsc_EpsUpdate_Active, p_EpsUpdate_TypeValue),
      oldGuti                    := p_MobileId,

      nonCurrentNativeNasKeySetId  := omit,

      gprsCipheringKeySeqNum     := cr_CiphKeySeqNum('8'H, ?) ifpresent,

      oldPtmsiSignature          := omit,

      additionalGuti             := omit,

      nonce                      := omit,

      ueNetworkCapability        := omit,

      lastVisitedRegisteredTai   := omit,

      drxParameter               := omit,

      ueRadioCapNeeded           := omit,

      epsBearerContextStatus     := omit,

      msNetworkCapability        := omit,

      oldLai                     := omit,

      tmsiStatus                 := omit,

      msClassmark2               := omit,

      msClassmark3               := omit,

      supportedCodecList         := omit,

      addUpdateType              := omit,

      voiceDomainPref            := omit,

      oldGUTI_Type               := omit,

      deviceProperties           := omit,

      msNetworkFeatureSupport    := omit,

      tmsiBasedNRIContainer      := omit,

      t3324Value                 := omit,

      t3412ExtdValue             := omit,

      extendedDRXParams          := omit,

      ueAddSecurityCap           := omit,

      ueStatus                   := omit,

      addInfoReq                 := omit

    }

  };


After Change

	template (present) NAS_UL_Message_Type cr_TAU_Request_S1ToN1(template (present)  EPS_UpdateTypeValue p_EpsUpdate_TypeValue,

                                                                NAS_KsiValue p_KsiValue,

                                                                template (present) MobileIdentity p_MobileId) :=

  { /* @status    APPROVED (NR5GC_IRAT) */

    tRACKING_AREA_UPDATE_REQUEST := {

      securityHeaderType         := tsc_SHT_NoSecurityProtection,

      protocolDiscriminator      := tsc_PD_EMM,

      messageType                := tsc_MT_TauRequest,

      asmeNasKeySetId            := cr_NAS_KeySetIdentifier(p_KsiValue, (tsc_NasKsi_NativeSecurityContext,tsc_NasKsi_MappedSecurityContext)),

      epsUpdateType              := cr_EPS_UpdateType(tsc_EpsUpdate_NotActive, p_EpsUpdate_TypeValue), /   

      oldGuti                    := p_MobileId,

      nonCurrentNativeNasKeySetId  := omit,

      gprsCipheringKeySeqNum     := cr_CiphKeySeqNum('8'H, ?) ifpresent,

      oldPtmsiSignature          := omit,

      additionalGuti             := omit,

      nonce                      := omit,

      ueNetworkCapability        := omit,

      lastVisitedRegisteredTai   := omit,

      drxParameter               := omit,

      ueRadioCapNeeded           := omit,

      epsBearerContextStatus     := omit,

      msNetworkCapability        := omit,

      oldLai                     := omit,

      tmsiStatus                 := omit,

      msClassmark2               := omit,

      msClassmark3               := omit,

      supportedCodecList         := omit,

      addUpdateType              := omit,

      voiceDomainPref            := omit,

      oldGUTI_Type               := omit,

      deviceProperties           := omit,

      msNetworkFeatureSupport    := omit,

      tmsiBasedNRIContainer      := omit,

      t3324Value                 := omit,

      t3412ExtdValue             := omit,

      extendedDRXParams          := omit,

      ueAddSecurityCap           := omit,

      ueStatus                   := omit,

      addInfoReq                 := omit

    }

  };


2.4 Change 4
	Function name
	f_EUTRA38_TrackingAreaUpdateFromN1

	Reason for change
	Following implementation of R5-202563, the function f_EUTRA38_TAUReqFromN1() to check the IE ueRadioCapNeeded based on “p_FirstN1_S1Transition”. This will currently always default to “true” as there is no provision to pass this information from f_EUTRA38_TrackingAreaUpdateFromN1().

	Summary of change
	Add parameter p_FirstN1ToS1 and pass to f_EUTRA38_TAUReqFromN1().
Note : This change has no affect in isolation because the parameter will still always default to true. Further CRs will be raised for test cases involving more tha 1 N1->S1 transition to explicitly set this parameter to false as required.

	TTCN module
	EUTRA38_NR5GCProcedures

	MCC160 Comment
	


Before Change

	  function f_EUTRA38_TrackingAreaUpdateFromN1(EUTRA_CellId_Type p_CellId,

                                              RRC_TransactionIdentifier p_RRC_TI,

                                              template EUTRA_CellId_Type p_CellIdPreviouslyCampedOn := eutra_Cell_NonSpecific,

                                              N1ToS1_SecurityContext_Type p_CxtType := MAPPED_CXT,

                                              RRCConnectionReleaseRequired_Type p_ReleaseRequired := rrcConnectionRelease,

                                              template (present) EstablishmentCause p_EstablishmentCause := mo_Signalling,

                                              template (omit) PLMN_List p_EPLMNlist := omit,

                                              template  MobileIdentity p_MobileId := omit) runs on EUTRA_5GS_PTC return NasCount_Type

{

    var NAS_MSG_Indication_Type v_NasInd;

    var template (present) MobileIdentity v_MobileId := cr_MobileIdAny;

    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(NR);

    var template (value) MobileIdentity v_MappedGUTI := f_EUTRA38_NR_Convert5G_CoOrdToEPS_GUTI (v_CoOrd_Guti, omit);

    if (isvalue(p_MobileId)) { // @sic R5s200564 sic@

      v_MobileId := p_MobileId;

    }

    if (p_CxtType == MAPPED_CXT) { // @sic R5s200564 sic@

      v_MobileId := v_MappedGUTI;

    }

    v_NasInd := f_EUTRA38_TAUReqFromN1(p_CellId,

                                       p_RRC_TI,

                                       v_MobileId,

                                       p_CellIdPreviouslyCampedOn,

                                       p_EstablishmentCause);
    return f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq(p_CellId,

                                                          p_RRC_TI,

                                                          v_NasInd,

                                                          p_CxtType,

                                                          p_ReleaseRequired,

                                                          p_EPLMNlist);

}


After Change

	function f_EUTRA38_TrackingAreaUpdateFromN1(EUTRA_CellId_Type p_CellId,

                                              RRC_TransactionIdentifier p_RRC_TI,

                                              template EUTRA_CellId_Type p_CellIdPreviouslyCampedOn := eutra_Cell_NonSpecific,

                                              N1ToS1_SecurityContext_Type p_CxtType := MAPPED_CXT,

                                              RRCConnectionReleaseRequired_Type p_ReleaseRequired := rrcConnectionRelease,

                                              template (present) EstablishmentCause p_EstablishmentCause := mo_Signalling,

                                              template (omit) PLMN_List p_EPLMNlist := omit,

                                              template  MobileIdentity p_MobileId := omit,

                                              boolean p_FirstN1ToS1 := true//@sic R5-202563 sic@  KS R5sxxx change_IRAT_common_5a

                                              ) runs on EUTRA_5GS_PTC return NasCount_Type

{

    var NAS_MSG_Indication_Type v_NasInd;

    var template (present) MobileIdentity v_MobileId := cr_MobileIdAny;

    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(NR);

    var template (value) MobileIdentity v_MappedGUTI := f_EUTRA38_NR_Convert5G_CoOrdToEPS_GUTI (v_CoOrd_Guti, omit);

    if (isvalue(p_MobileId)) { // @sic R5s200564 sic@

      v_MobileId := p_MobileId;

    }

    if (p_CxtType == MAPPED_CXT) { // @sic R5s200564 sic@

      v_MobileId := v_MappedGUTI;

    }

    v_NasInd := f_EUTRA38_TAUReqFromN1(p_CellId,

                                       p_RRC_TI,

                                       v_MobileId,

                                       p_CellIdPreviouslyCampedOn,

                                       p_EstablishmentCause,

                                       p_FirstN1ToS1);//@sic R5-202563 sic@  KS R5sxxx change_IRAT_common_5b

    return f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq(p_CellId,

                                                          p_RRC_TI,

                                                          v_NasInd,

                                                          p_CxtType,

                                                          p_ReleaseRequired,

                                                          p_EPLMNlist);

}


…

